TEMPLATE = lib

#Application version
#VERSION = 0.4.0

CONFIG += warn_on thread exceptions rtti stl
CONFIG += debug_and_release warn_on


CONFIG(debug, debug|release) {
    TARGET = qBreakpad_d
} else {
    TARGET = qBreakpad
}

message(target $$TARGET)

QT -= gui
QT += core network

OBJECTS_DIR = $$PWD/_build/obj
MOC_DIR = $$PWD/_build
DESTDIR = $$PWD


### qBreakpad config
include(config.pri)

## google-breakpad
include(third_party/breakpad.pri)

HEADERS += \
    singletone/call_once.h \
    singletone/breakpad_singleton.h \
    QBreakpadHandler.h \
    QBreakpadHttpUploader.h

SOURCES += \
    QBreakpadHandler.cpp \
    QBreakpadHttpUploader.cpp

mac:{
    QMAKE_MACOSX_DEPLOYMENT_TARGET = 10.10
}
